# Maintainer: Felix Yan <felixonmars@archlinux.org>

pkgname=deepin-album
pkgver=6.0.0
pkgrel=2
pkgdesc='A fashion photo manager for viewing and organizing pictures'
arch=('loong64' 'x86_64')
url="https://github.com/linuxdeepin/deepin-album"
license=('GPL3')
depends=('deepin-qt5integration' 'dtkdeclarative' 'ffmpeg' 'ffmpegthumbnailer' 'gio-qt'
         'udisks2-qt5')
makedepends=('cmake' 'ninja' 'qt5-tools')
groups=('deepin-extra')
source=("$pkgname-$pkgver.tar.gz::https://github.com/linuxdeepin/deepin-album/archive/$pkgver.tar.gz")
sha512sums=('e67cf30531ce9e37d8ee6765270da4620b019656ac825919fa34f79a62bc6f59de4d0066a51dce6f89f7123b73471b8351848a5c0799e51683f7893f40f364fa')
options=(!lto)

build() {
  cd deepin-album-$pkgver
  cmake . -GNinja -DCMAKE_INSTALL_PREFIX=/usr -DVERSION=$pkgver
  ninja
}

package() {
  cd deepin-album-$pkgver
  DESTDIR="$pkgdir" ninja install
}
